
ELEC 379 : DESIGN OF DIGITAL AND M ICROCOMPUTER SYSTEMS

1998/99 WINTER SESSION, TERM 2

PCInterruptStructureand8259DMA Controllers
This lecture covers the use of interrupts and the vectored interrupt mechanism used on the IBM PC using the Intel
8259 Programmable Interrupt Controller (PIC).
After this lecture you should be able to: (1) decide and explain why interrupts should or should not be used to service
a particular peripheral, (2) describe how the 8259 PIC handles multiple interrupt sources, and (3) write an ISR in
8088 assembly language to service interrupts generated by the 8259 PIC.

Overview of I/O Strategies

WhendesigninganI/O interfacewe mustmake two
fundamentaldecisions:(1) how we determinethat
a peripheralis readyto acceptor supply ore data,
and (2) how the datais to be transferredfrom the
peripheralto the computer’s memory. For eachof
thesedecisionstherearetwo commonapproaches.

To decidewhenwecantransferdatato/fromade-
vicewehavetwochoices.Thesimpleschoice,called
polling, is to periodicallycheckastatusregister. The
secondchoice,calledinterrupt-driven I/O, is to con-
nectasignalontheperipheralto theCPU’s interrupt
input. In this casewhenthe I/O device is readyto
transferdatatheexecutionof whateverprogramhap-
pensto berunningis temporarilysuspendedandthe
CPUrunsanotherprogram,aninterruptservicerou-
tine (ISR) thatperformstheI/O operations.

To actually transferthe datawe againhave two
choices. The simplestchoice,programmed I/O, is
for the processorto read/writethe data to/from a
dataregister on the peripheraland write/readit to
memory. Anotherpossibility, Direct Memory Access
(DMA), is to make useof hardwarethat“takesover”
the CPU bus andtransfersthe datadirectly from/to
theperipheralto/frommemory.

Thetablebelow summarizesthefour choices:

DataTransfer
When How

Simple: Polling ProgrammedI/O
Complex: Interrupts DMA

In this lecturewe cover the designof interrupt-
driver I/O deviceson IBM PC compatiblearchitec-
tures.

Interrupts versus Polling

I/O devicessuchasprinters,keyboards,etc. require
that the CPU executespecialcodeto “service” the
deviceeveryoncein awhile. For example,incoming
charactersor keystrokeshave to bereadfrom a data
registeron theperipheralandstoredin abuffer to be
usedlaterby theoperatingsystem.

Thetwo commonwaysof servicingdevicesareby
polling andby using interrupts. Polling meansthe
statusof theperipheralis checkedperiodicallyto de-
terminewhetherit needsto beserviced,for example
whetherthedevicehasdatareadyto beread.Theal-
ternativeis to useinterrupts. Theperipheralinterface
is designedto asserttheinterruptrequestinput to the
CPUwhenit requiresservice.Theresultof asserting
theinterruptsignalis to interruptnormalflow of con-
trol andto causeaninterruptserviceroutine(ISR)to
beexecutedto servicethedevice.

Polling mustbe donesufficiently fastthat datais
not lost. For example,if aserialinterfacecanreceive
up to 1000characterspersecondandcanonly store
thelastcharacterreceived,it mustbecheckedatleast
oncepermillisecondto avoid losingdata.Sincewe
needto periodicallycheckeachdevice,regardlessof
whetherit requiresserviceor not, polling causesa
fixedoverheadfor eachinstalleddevice.

Another, possiblygreater, disadvantageof polling
is that polling routinesmustbe integratedinto each
andeveryprogramthatwill usethatperipheral.Pro-
gramsmustbe written to periodicallypoll andser-
viceall theperipheralsthey use.Suchtight coupling
betweentheapplicationandthehardwareis usually
undesirableexceptin thesimplestembeddedproces-
sorcontrolsystems.

On theotherhand,an ISR is only executedwhen
adevice requiresattention(e.g.a characterhasbeen
received). Thusthereis no fixedoverheadfor using

lec10.tex 1



interrupt-driven devices. In addition,sinceISRsop-
erateasynchronouslywith theexecutionof otherpro-
grams,� it is notnecessaryfor applicationprogramsto
worry aboutthedetailsof the I/O devices. An ISRs
is usuallyprovided aspart of a device driver which
is usuallyincludedaspartof anoperatingsystem.

However, respondingto an interrupttypically re-
quiresexecutingadditionalclock cyclesto save the
processorstate,fetch the interruptnumberand the
correspondinginterruptvector, branchto theISRand
laterrestoretheprocessorstate.

Someto considerwhen decidingwhetherto use
polling or interruptsto decidewhento servicea de-
vice include:

� Canthedevicegenerateinterrupts?If thedevice
or the CPU arevery simplethenthey may not
have beendesignedto generateor respondto
interrupts.

� How complex is theapplicationsoftware?If the
applicationis acomplex programthatwouldbe
difficult to modify in orderto addperiodicpolls
of thehardwarethenyoumayhave to useinter-
rupts. On the otherhand,if the applicationis
a controllerthatsimply monitorssomesensors
andcontrolssomeactuatorsthenpollingmaybe
thebestapproach.

� What is the maximum time allowed between
polls? If the device needsto be servicedwith
very little delaythenit maynot bepracticalto
usepolling.

� What fractionof polls areexpectedto resultin
datatransfer?If therateat which thedevice is
polled is much higher than the averagetrans-
fer rate then a large fraction of polls will be
“wasted”and using interruptswill reducethis
polling overhead.

In general,useinterruptswhenthe overheaddue
to polling would consumea large percentageof the
CPUtimeor wouldcomplicatethedesignof thesoft-
ware.

Exercise: Data is arriving on a serial interface at 4000 char-

acters per second. If this device is serviced by polling, and each

character must be read before another one is received, what is

the maximum time allowed between polls? If each poll requires

10 microseconds to complete, what fraction of the CPU time is

always being used up even when the serial port is idle? What if

there were 8 similar devices installed in the computer?

Exercise: Data is being read from a tape drive interface at

100,000 characters per second. The overhead to service an in-

terrupt and return control to the interrupted program is 20 mi-

croseconds. Can this device use an ISR to transfer each charac-

ter?

Exercise: Responding to an interrupt typically takes consid-

erably longer than polling a status bit. Why are interrupts useful?

Alternative Approaches

It’s also possibleto usea mixture of interrupt and
polleddevices. For example,a device canbepolled
by an ISR that executesperiodicallydueto a clock
interrupt. This removesthe needto includepolling
routinesin eachapplication. We canalsopoll sev-
eral devices in an ISR that runs periodically. This
maybemoreefficient thathaving eachdevice issue
independentinterrupts.

It is alsocommonfor devices to buffer multiple
bytesandissuean interruptonly whenthe buffer is
full (or empty). TheISR canthentransferthecom-
pletebuffer without incurringtheinterruptoverhead
for eachbyte. For example,modemPCserialinter-
facescanstoreupto 16bytesbeforeissuinganinter-
rupt. Thiscutsdown theinterruptoverheadby up to
16.

Becauseinterruptsoccurdueto eventsoutsidethe
computer’s control, it is usually difficult to predict
theexactsequenceandratein which interruptswill
happen.In applicationswherelossof dataabsolutely
cannotbe tolerated(e.g. wheresafetyis a concern)
thedesignermustensurethatall of the devicesser-
viced by interruptscan be properlyservicedunder
the worst-caseconditions. Typically this involvesa
sequenceof nestedinterruptshappeningcloselyone
afteranotherin a particularorder. In someof these
systemsit maybeeasierto usepolling to ensurecor-
rectworst-casebehaviour.

Exercise: Consider a monitoring system in a nuclear power

plant. The system is hooked up to hundreds of sensors, each

of which can indicate an error condition. It is difficult to predict

exactly how often and in what order these error conditions will

happen. Would you design the system so that alarm conditions

generated interrupts? Why or why not?

2



Maskable, Non-Maskable and Soft-
ware Interrupts

Likemany otherprocessors,the80386hastwo types
of interrupts: maskableand non-maskable.Mask-
ableinterrupts(assertedontheINTR pin) canbedis-
abledby clearingtheinterrupt-enableflag (IF bit) in
theflagsregisterusingtheCLI instruction.A mask-
ableinterruptcausesaninterruptacknowledgecycle
(similar to a readcycle) which readsa 1-byteinter-
rupt type from the interruptingperipheral. The in-
terrupt type (which is not the sameasthe interrupt
“number”)is thenmultipliedby four andaninterrupt
vectoris fetchedfrom thisaddress.

Non-maskableinterrupts (assertedon the NMI
pin) cannotbe disabled.ThusNMI is usuallyused
for very high priority eventssuchas imminent loss
of power or a hardware fault. A NMI alwaysuses
the interruptvectorfor interrupttype2, thusallow-
ing it executefaster. For example,on thePCNMI is
assertedif thehardwarediscoversamemoryerror.

Software interruptsoperatein the sameway as
maskableandnon-maskableinterruptsbut they are
generatedby executinganINT instruction.Theinter-
rupt typeis suppliedin theinstructionandso,again,
no interruptacknowledgecycle is required.

In addition, certainerror conditions(suchas di-
vide by zero)cancauseexceptions which behave in
thesamewayassoftwareinterrupts.

Exercise: How could you execute the NMI handler on a PC if

you had to regain control after it executed? How else could you

do this if the NMI ISR did not return control?

Exercise: In “real mode” each 386 interrupt vector requires

4 bytes. What is the maximum number of bytes used up by an

interrupt vector table?

Interrupt Processing

The following sequenceof events happensin re-
sponseto aninterrupt:

1. thecurrentinstructionis completed

2. andinterruptacknowledgecycle is run andthe
CPUreadsaninterrupttypefrom thehardware

3. the CPU saves the processorcontext (flags,
IP and CS registersare pushedon the current
stack)

4. theinterrupt-enableflag(IF) is cleared

5. an interruptvector (the locationof the ISR) is
retrieved from the interruptvector tableat the
memoryaddressgivenby the interruptnumber
multipliedby 4

6. theCPUbeginsexecutionof theISR

Thefirst two stepsareskippedin thecaseof NMI
andsoftwareinterrupts.

The 8259 in the IBM PC Architecture

The 80386CPU only hasoneinterruptrequestpin.
Although simplesystemsmay only have one inter-
rupt source,mostsystemsmust have someway of
dealingwith multiple interrupt sources. The Intel
“way of doing things” is to usea chip calleda pro-
grammableinterruptcontroller(PIC).Thischiptakes
as inputs interrupt requestsignalsfrom up to 8 pe-
ripheralsand suppliesa single INTR signal to the
CPUasshown below:

ThePIChas3 purposes:

1. It allows eachof the individual interruptsto be
enabledor disabled(masked).

2. It prioritizesinterruptssothat if multiple inter-
ruptshappenat thesametime theonewith the
highestpriority is servicedfirst. The priorities
of the interruptsarefixed,with input IR0 hav-
ing thehighestpriority andIR7 the lowest. In-
terruptsof alowerpriority nothandledwhile an
ISR for ahigher-level interruptis active.

3. It providesaninterrupttypethattheCPUreads
during the interrupt acknowledgecycle. This
tells the CPU which of the 8 possibleinter-
ruptsoccurred.ThePIC on theIBM PCis pro-
grammedto respondwith aninterrupttypeof 8

plus the particularinterruptsignal(e.g. if IR3
wasassertedtheCPUwould readthe value11
from thePIC duringthe interruptacknowledge
cycle).

Thefollowing diagramshows how eachof thein-
terruptrequestlinesto thePIC canpotentiallycause
aninterruptrequestto bemadeto theCPU.TheCPU
readsthe interrupttype from the PIC during the in-
terruptacknowledgecycle andthenusesthis typeto
look up theaddressof theISR in theinterruptvector
table.

interrupt
number

interrupt
   type

interrupt
vector
table

address
    of
   ISR

CPU

data
bus

IRx

INTR

PIC

INT

Unlike many othermicroprocessorsbothINT and
IRx areactive-highsignals.On theIBM PCtheIRx
inputsareconfiguredto beedge-triggered.Thesede-
signchoicesprevent thesharingof interruptrequest
lines.

Oncethe PIC hasbeenconfigured,therearetwo
control registersthatcanbe reador written. On the
IBM PC and compatiblemachinesthe addressde-
coderfor PIC placesthesetwo registersin the I/O
(not memory) addressspaceat locations20H and
21H.

On the IBM AT andlater modelstherearemore
than8 interruptsourcesandtherearetwo PICs.The
slave PIC supportsan additional8 interrupt inputs
andrequestsaninterruptfrom themasterPICasif it
wereaninterruptingperipheralon IR2.

Exercise: What is the maximum number of interrupt sources

that could be handled using one master and multiple slave

PICs?

Exercise: Compare this approach to that used for vectored

interrupts on typical 68000 systems. How many interrupt sources

can be connected directly to a 68000? What if a priority encoder

is used? Are interrupt request lines active-high or active-low?

What device supplies the interrupt number or interrupt vector in

a typical 68000 system? After the interrupt is serviced a typical

68000 peripheral will un-assert it’s interrupt request output. How

does an 80x86 system determine that the device no longer needs

attention?

Interrupt Number and Interrupt
Type

A commonsourceof confusionis thedifferencebe-
tweentheinterruptnumber, whichis theinterruptre-
questpin thatis assertedandtheinterrupttypewhich
is thevaluereadby theCPUduringtheinterruptac-
knowledgecycleor suppliedin anINT instruction.

The interrupt inputs to the PIC areconnectedas
follows ona IBM PC-compatiblesystem:

interrupt device
0 timer
1 keyboard
2 reserved
3 serialport2
4 serialport1
5 harddisk
6 floppy disk
7 printer1

Thefollowing aresomeof theinterrupttypesthat
arepre-definedon80x86CPUs:

interrupttype cause
0 Divideby Zero
1 SingleStep
2 NMI
3 Breakpoint
4 Overflow

� �

8 to 255 implementation-dependent

Note that thesearenot the sameas the interrupt
numbers.

Exercise: On an IBM PC-compatible system what interrupt

number is used for a floppy-disk interrupt? What interrupt type

will the CPU see for this interrupt? At what addresses will the

CPU find the interrupt vector for this interrupt?

Exercise: When the a key on the keyboard is pressed, which

input on the 8259 PIC will be asserted? What will the signal

level be? What value will the 80386 read from the PIC during the

interrupt acknowledge cycle?

Programming the 8259 Interrupt
Controller

The initialization of the PIC is rathercomplicated
becauseit hasmany possibleoperatingmodes.The

PIC’s operatingmodeis normally initialized by the
BIOS whenthesystemis booted.We will only con-
siderthestandardPIC operatingmodesusedon the
IBM PC and only a systemwith a single (master)
PIC.

In it’s standardmodethePICoperatesasfollows:

� if no ISRfor thesameor ahigherlevel is active
the interruptrequest(INTR) signalto theCPU
is asserted

� if theCPU’s interruptenableflag is setthenan
interruptacknowledgecycle will happenwhen
thecurrentinstructionterminates

� during the interrupt acknowledge cycle the
highest-priority interrupt request is captured
andsaved (“latched”) in thePIC’s interruptre-
questregister(IRR) andthenthe interrupttype
for this interrupt is readby the CPU from the
PIC. An interrupt acknowledgeactually takes
two clockcycles.

TheCPUusesthe interrupttypeto look up the
addressof theISR andrunsit

� at the end of the ISR a commandbyte (20H)
must be written to the PIC register at address
20H to re-enableinterruptsat that level again.
This is calledthe‘EOI’ (end-ofinterrupt)com-
mand.

Duringnormaloperationonly two operationsneed
to beperformedon thePIC:

1. Disabling (masking) and enabling interrupts
from a particularsource.This is doneby read-
ing the interruptmaskregister(IMR) from lo-
cation21H,usinganAND or ORinstructionto
set/clearparticularinterruptmaskbits.

2. Re-enablinginterrupts for a particular level
when the ISR for that level complete. This
is done with the EOI commandas described
above.

Masking/Enabling Interrupts

Thereare threeplaceswhereinterruptscanbe dis-
abled:(1) thePICinterruptmask,(2) thePICpriority
logic, and(3) theCPU’s interruptenableflag.

Exercise: What is the difference between an interrupt “mask”

bit and an interrupt “enable” bit?

If thePIC interruptmaskbit is setthenthe inter-
rupt requestwill not be recognized(or latched). If
thePICbelievesanISRfor ahigherlevel interruptis
still executingdueto no EOI commandhaving been
given for that interrupt level it will not allow inter-
ruptsof thesameor lower levels. If theinterrupten-
ablebit in theCPU’sflagsregisteris notsetthenthe
interruptrequestsignalfrom thePICwill beignored.

NotethattheCPU’sinterruptenableflagis cleared
whenan interrupthappensandis restoredwhenthe
processreturnsfrom the ISR via the IRET instruc-
tion. This meansthat ISRscan’t be interrupted(not
evenby ahigher-level interrupt)unlessinterruptsare
explicitly re-enabledin theISR.

It’s possibleto allow theCPUto interruptanISR
(resultingin nested interrupts) by settingthe inter-
ruptenablebit with theSTI instruction.

Exercise: Can interrupts on an IBM-PC compatible computer

be nested (i.e. can an ISR be interrupted)? If so, under what

conditions?

Exercise: How many levels deep can interrupts be nested

on the IBM PC if the ISR does not re-enable interrupts? If it re-

enables interrupts but does not issue EOI to the PIC? If it does

both? In each of these cases how much space would be required

on the interrupted program’s stack to hold the values pushed dur-

ing the interrupt acknowledge cycle if 8 bytes are saved during

each interrupt?

Interrupt Latency

Oftena peripheralmustbeservicedwithin a certain
time limit after an event. For example,a character
mustbe readfrom an input port beforethe next on
arrives.

The interruptlatency is the maximumtime taken
to respondto an interruptrequest.This will include
the time it takesfor the currentinstructionto com-
pleteplusthetime for theCPUto respondto thein-
terrupt(e.g. save the CS, EIP andflag registerson
thestack,acknowledgetheinterruptandfetchthein-
terruptvector). If an ISR is alreadyexecutingand
cannotbeinterruptedthenthis alsoincreasesthein-
terruptlatency.

Interruptroutinesshouldbekeptasshortaspossi-
ble to minimize the interruptlatency. Typically this

involves having the ISR storevaluesin a buffer or
setflagsandthenhaving the bulk of the processing
performed� outsidetheISR.A typical “devicedriver”
consistsof an ISR that executesonly time-critical
functionssuchas reading/writingdata from/to the
peripheralandanotherportionthatdealswith higher-
level issuessuch as moving the disk drive head,
checkingfor errors,etc.

Edge- and Level-Triggered Inter-
rupts

Interruptrequestsignalscanbedesignedto beedge-
triggered(theinterruptactsasa clock andtherising
(or falling) edgeof theinterruptsignalcausesanin-
terrupt to be recorded)or level-triggered(the inter-
rupt controllersamplestheinterruptsignalat certain
timesandrecordsaninterruptif theinput is asserted.

Exercise: The 8259 PIC is configured for edge-triggered in-

terrupts. Is it possible to share the interrupt request inputs by

wire-OR’ing several interrupt sources? Why or why not? What if

the inputs were active-low?

Sample 80386/8259 ISR

Thecodebelow shows an80386assemblylanguage
programthat includesan ISR. The programsetsup
anISRfor interruptnumber8 (thetimer interrupton
the IBM PC). The ISR simply decrementsa count.
Themainprogramwaitsuntil thecountreacheszero
andthenterminates.

Thetimer interrupton the IBM PCis drivenby a
clock thatgeneratesoneinterruptevery 55 millisec-
onds.With theinitial countvalueprovidedbelow the
programwaitsfor 15secondsbeforeterminating.

Themainprogramsavesandrestorestheprevious
timer interruptvector.

When the ISR begins executiononly the IP and
CS registerswill have beeninitialized. Any other
segmentregistersthat will be usedin the ISR must
be explicitly loaded. In this casethe codeanddata
areasarelocatedin thesamesegmentsoDS canbe
loadedfrom CS.

On entry to the ISR only the IP, CS and flags
registerswill have beensaved on the caller’s stack.
Any other registers used by the ISR must be saved

when starting the ISR and restored before return-
ing. Otherwisethestateof the interruptedcodewill
be changedby the ISR and this is likely to cause
seemingly-randomfailuresin otherprograms.

;
; example of program using an ISR for
; IBM PC timer interrupt
;

isrvec equ 4*(8+0) ; location of vector for IR0

code segment public ; .COM file setup
assume cs:code,ds:code
org 100h

start:
mov ax,0 ; use ExtraSegment to access
mov es,ax ; vectors in segment 0

; save old interrupt vector

mov ax,es:[isrvec]
mov prevoff,ax
mov ax,es:[isrvec+2]
mov prevseg,ax

; set up new vector

cli ; disable interrupts until
; vector update is complete

mov ax,offset isr
mov es:[isrvec],ax
mov ax,cs
mov es:[isrvec+2],ax

sti ; re-enable interrupts

; wait until ISR decrements count to zero

loop: mov ax,count
cmp ax,0
jnz loop

; restore old interrupt vector

cli ; disable interrupts until
; vector update is complete

mov ax,prevoff ; restore prev.
mov es:[isrvec],ax ; offset/segment
mov ax,prevseg
mov es:[isrvec+2],ax

sti ; re-enable
; interrupts

; return to DOS

int 20h

; storage for demonstration program

count dw 273
prevoff dw ?

prevseg dw ?

; The ISR itself:

isr:
mov cs:tmpax,ax ; save working registers
mov ax,ds
mov cs:tmpds,ax

mov ax,cs ; set up DS
mov ds,ax

mov ax,count
cmp ax,0 ; don’t decrement if already zero
jz isr1
sub ax,1 ; decrement count
mov count,ax

isr1:

mov al,20h ; write EOI command to 8259 PIC
out 20h,al ; to re-enable interrupts

mov ax,tmpds ; restore working registers
mov ds,ax
mov ax,cs:tmpax

iret ; return from ISR and
; re-enable interrupts

tmpax dw ?
tmpds dw ?

code ends
end start

Exercise: Why must interrupts be disabled while updating the

interrupt vector?

Exercise: How will the PC’s time of day change when this

program is run? What would happen if the interrupt were not

restored?

Exercise: Could a stack be used to save the values of the

registers that will be changed in the ISR? Which stack? What

are the advantages and disadvantages of doing so?

